
Specifying Syntax

Language Specification

Syntax

Form of phrases

Physical arrangement of symbols

Semantics

Meaning of phrases

Result of executing a program

Pragmatics

Usage of the language

Features of implementations

Components of a Grammar

1. Terminal symbols or terminals, Σ

2. Nonterminal symbols or syntactic
categories, N

Vocabulary = Σ ∪ N

3. Productions or rules, α  ::=  β,

where α,β∈ (Σ ∪ N)*

4 .Start Symbol   (a nonterminal)

Types of Grammars

Type 0:Unrestricted grammars

α  ::=  β
where α contains a nonterminal

Type 1:Context-sensitive grammars

αBγ ::= αβγ

Type 2:Context-free grammars

A ::= string of vocabulary symbols

Type 3:Regular grammars

A ::= a   or   A ::= aB

An English Grammar

<sentence> ::= <noun phrase> <verb phrase> .

<noun phrase> ::= <determiner> <noun>
| <determiner> <noun> <prepositional phrase>

<verb phrase> ::= <verb> | <verb> <noun phrase>
| <verb> <noun phrase> <prepositional phrase>

<prepositional phrase> ::=
<preposition> <noun phrase>

<noun> ::= boy | girl | cat | telescope
| song | feather

<determiner> ::= a | the

 <verb> ::= saw | touched | surprised | sang

 <preposition> ::= by | with

A Derivation (for Figure 1.4)

 <sentence>  ⇒  <noun phrase> <verb phrase> .
⇒  <determiner> <noun> <verb phrase> .
⇒  the <noun> <verb phrase> .
⇒  the girl <verb phrase> .
⇒  the girl <verb> <noun phrase> .
⇒  the girl touched <noun phrase> .
⇒  the girl touched <determiner>

<noun> <prep phrase> .
⇒  the girl touched the <noun> <prep phrase> .
⇒  the girl touched the cat <prep phrase> .
⇒  the girl touched the cat <preposition>

<noun phrase> .
⇒  the girl touched the cat with <noun phrase> .
⇒  the girl touched the cat with

<determiner> <noun> .
⇒  the girl touched the cat with a <noun> .
⇒  the girl touched the cat with a feather .

Note example of a context-sensitive grammar.

Chapter 1 6

Concrete Syntax for Wren

Phrase-structure syntax:

<program> ::= program <ident> is <block>

<block> ::= <declaration seq>
begin <cmd seq> end

<declaration seq> ::= ε
| <declaration> <declaration seq>

<declaration> ::= var <var list> : <type> ;

<type> ::= integer | boolean

<var list> ::= <var> | <var> , <var list>

<cmd seq> ::= <cmd> | <cmd> ; <cmd seq>

<cmd> ::= <var> := <expr> | skip

| read <var> | write <int expr>

| while <bool expr> do <cmd seq> end while

| if <bool expr> then <cmd seq> end if

| if <bool expr> then <cmd seq>
else <cmd seq> end if

<expr> ::= <int expr> | <bool expr>

<integer expr> ::= <term>
| <int expr> <weak op> <term>

<term> ::= <element>
| <term> <strong op> <element>

<element> ::= <numeral> | <var>
| ( <int expr> ) | – <element>

<bool expr> ::= <bool term>
| <bool expr> or <bool term>

<bool term> ::= <bool element>
| <bool term> and <bool element>

<bool element> ::= true | false | <var>
| <comparison>
| not ( <bool expr> ) | ( <bool expr> )

<comparison> ::= <int expr> <relation> <int expr>

<var> ::= <ident>

Lexical syntax:

<relation> ::= <= | < | = | > | >= | <>

<weak op> ::= + | – <strong op>  ::=  * | /

<ident> ::= <letter> | <ident> <letter>
| <ident> <digit>

<letter> ::= a | b | c | d | e | f | g | h | i | j | k | l | m
    | n | o | p | q | r | s | t | u | v | w | x | y | z

<numeral> ::= <digit> | <digit> <numeral>

<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

Tokens:

<token> ::= <identifier>  |  <numeral>
|  <reserved word>  |  <relation>
|  <weak op>  |  <strong op>
|  :=  |  (  |  )  |  ,  |  ;  |  :

<reserved word> ::= program | is | begin | end
| var | integer | boolean | read
| write | skip | while | do | if | then
| else | and | or | true | false | not

Sample Wren Program

program prime is
var num,divisor : integer;
var done : boolean;

begin
read num;
while num>0 do

divisor := 2;   done := false;
while divisor<= num/2 and not(done) do

done := num = divisor*(num/divisor);
divisor := divisor+1

  end while;
if done then write 0

else write num
 end if;
read num

  end while
end

Ambiguity

One phrase has two different derivation trees,
thus two different syntactic structures.

Ambiguous structure may lead to ambiguous
semantics.

Dangling “else”

Associativity of expressions (ex 2, page 16)

Ada:
Is “a(2)” a subscripted variable

or a function call?

Classifying Errors in Wren
Context-free syntax

— specified by concrete syntax
• lexical syntax
• phrase-structure syntax

Context-sensitive syntax (context constraints)
• also called “static semantics”

Semantic or dynamic errors

Context Constraints in Wren
1. The program name identifier may not be

declared elsewhere in the program.

2. All identifiers that appear in a block must be
declared in that block.

3. No identifier may be declared more than once
in a block.

4. The identifier on the left side of an
assignment command must be declared
as a variable, and the expression on the
right must be of the same type.

5. An identifier occurring as an (integer)
element must be an integer variable.

6. An identifier occurring as a Boolean element
must be a Boolean variable.

7. An identifier occurring in a read command
must be an integer variable.

<token>*  —  abitrary strings of tokens

<program>  —  strings derived from 
                         the context-free grammar

Syntactically well formed 
Wren programs derived from 
<program> and also satisfying 
the context constraints

Syntax of Wren

Semantic Errors in Wren
1. An attempt is made to divide by zero.

2. A variable that has not been initialized is
accessed.

3. A read command is executed when input file
empty.

4. An iteration command (while) does not
terminate.

Regular Expressions Denote Languages

∅

ε

a for each a∈Σ

(E | F) union

(E • F) concatenation

(E*) Kleene closure

May omit some parentheses following the
precedence rules

Abbreviations

E1 E2 = E1 • E2

E+ = E • E*

En = E • E • … • E,  repeated n times

E? = ε | E

Note:  E* = ε | E1 | E2 | E3 | …

Nonessential Ambiguity
Command Sequences:

<cmd seq> ::= <command>
| <command> ; <cmd seq>

<cmd seq> ::= <command>
| <cmd seq> ; <command>

<cmd seq> ::= <command> ( ; <command> )*

Or include command sequences with
commands:

<command> ::= <command> ; <command>
| skip | read <var> | …

Language Processing:

scan : Character* → Token*
parse : Token* → ????

Concrete Syntax (BNF)

• Specifies the physical form of programs.

• Guides the parser in determining the phrase
structure of a program.

• Should not be ambiguous.

• A derivation proceeds by replacing one
nonterminal at a time by its corresponding
right-hand side in some production for that
nonterminal.

• A derivation following a BNF definition of a
language can be summarized in a derivation
(parse) tree.

• Although a parser may not produce a
derivation tree, the structure of the tree is
embodied in the parsing process.

Chapter 1 16

read m; if m=-9 then m:=0 end if; write 22*m

Derivation Tree

<str op>

<cmd seq><bool expr>

<rel>

<term>

<expr>

=

<int expr>

<element>

num(9)

var(m)

if then end if

<bool term>

<bool elem>

<comparison>

<term>

<int expr>

<element>

;<command> <cmd seq>

<command>

read

<term>

var(m)

<variable>

var(m)

:=

<int expr>

num(22)

<elem><term>

<element> var(m)

;<command> <cmd seq>

<command>

<variable>

write <int expr>

num(0)

<term>

<element>

*

<cmd seq>

- <element>

Abstract Syntax

• Representation of the structure of language
constructs as determined by the parser.

• The hierarchical structure of language
constructs can be represented as abstract
syntax trees.

• The goal of abstract syntax is to reduce
language constructs to their “essential”
properties.

• Specifying the abstract syntax of a language
means providing the possible templates for the
various constructs of the language.

• Abstract syntax generally allows ambiguity
since parsing has already been done.

• Semantic specifications based on abstract
syntax are much simpler.

read m; if m=-9 then m:=0 end if; write 22*m

Abstract Syntax Tree:

O R

times num(22)

expr

var(m)

write

expr

num(9)

equal var(m)

if

minus

if

times

assign

commands

var(m)

read write

equal

var(m) minus

commands

var(m) num(0)

num(22) var(m)

num(9)

Designing Abstract Syntax

Consider concrete syntax for expressions:

<expr> ::= <int expr> | <bool expr>

<integer expr> ::= <term>
| <int expr> <weak op> <term>

<term> ::= <element>
| <term> <strong op> <element>

<element> ::= <numeral> | <var>
| ( <int expr> ) | – <element>

<bool expr> ::= <bool term>
| <bool expr> or <bool term>

<bool term> ::= <bool element>
| <bool term> and <bool element>

<bool element> ::= true | false | <var>
| <comparison> | ( <bool expr> )
| not ( <bool expr> )

<comparison> ::=
<int expr> <relation> <int expr>

Assume we only have to deal with programs that
are syntactically correct.

What basic forms can an expression take?

<int expr> <weak op> <term>

<term> <strong op> <element>

<numeral> <variable>

– <element>

<bool expr> or <bool term>

<bool term> and <bool element>

true false
not ( <bool expr> )
<int expr> <relation> <int expr>

Abstract Syntax:

Expression ::= Numeral | Identifier | true | false
| Expression Operator Expression
| – Expression | not(Expression)

Operator ::= + | – | * | / | or | and
  | <= | < | = | > | >= | <>

Abstract Syntax
Abstract Syntactic Categories

Program Type Operator
Block Command Numeral
Declaration Expression Identifier

Abstract Production Rules

Program ::= program Identifier is Block

Block ::= Declaration* begin Command+ end

Declaration ::= var Identifier+ : Type ;

Type ::= integer | boolean

Command ::= Identifier := Expression | skip
| read  Identifier | write Expression
| while Expression do Command+

| if Expression then Command+

| if Expression then Command+ else Command+

Expression ::= Numeral | Identifier | true | false
| Expression Operator Expression
| – Expression | not( Expression)

Operator ::= + | – | * | / | or | and
| <= | < | = | > | >= | <>

Alternative Abstract Syntax

 Abstract Production Rules

Program ::= prog (Identifier, Block)

Block ::= block (Declaration*, Command+)

Declaration ::= dec (Identifier+, Type)

Type ::= integer | boolean

Command ::= assign (Identifier, Expression)
| skip  | read (Identifier) | write (Expression)
| while (Expression, Command+)
| if (Expression, Command+)
| ifelse (Expression, Command+, Command+)

Expression ::= Numeral | Identifier | true | false
| not (Expression) | minus (Expression)
| exp (Operator, Expression, Expression)

Operator ::= + | – | * | / | or | and
| <= | < | = | > | >= | <>

Example

read m;
if m=-9 then m:=0 end if;

write 22*m

[ read (m),

if (exp (equal, ide(m),minus (num(9))),

[ assign (m, num(0)) ]),

write (exp (times, num(22), ide(m)))

]

Language Processing

scan : Character* → Token*

parse : Token* → Abstract Syntax Tree

Are either of these functions one-to-one?
Explain.


